
Utiliser LibreOffice sur un Mac

Sur Mac, certaines touches et certains éléments de menu sont différents de ceux utilisés sous 
Windows ou Linux. Le tableau ci-dessous donne quelques substitutions courantes pour les instruc-
tions de ce chapitre. Pour une liste plus détaillée, voyez l’Aide de l’application.

Windows ou Linux Équivalent Mac Effet

Sélection du menu 
Outils > Options

LibreOffice > Préférences Accès aux options de configuration

Clic droit
Control+clic ou clic droit 
selon la configuration de l’ordinateur

Ouvre un menu contextuel

Ctrl (Control) ⌘ (Command) Utilisé avec d’autres touches

F5 Shift+⌘+F5 Ouvre le navigateur

F11 ⌘+T
Ouvre l’onglet Styles et Formatage 
du volet latéral

La documentation française de LibreOffice est disponible à
https://wiki.documentfoundation.org/Documentation/Publications/fr

Introduction

Dans les chapitres précédents, nous avons saisi dans chaque cellule l’un des deux types de don-
nées de base : les nombres et les textes. Cependant, nous ne savons pas toujours ce que le 
contenu devrait être. Souvent, le contenu d’une cellule dépend du contenu d’autres cellules. Pour 
gérer cette situation, nous allons utiliser un troisième type de données : les formules. Les formules 
sont des équations qui utilisent des nombres, des textes et des variables pour obtenir un résultat. 
Dans un classeur, les variables sont des emplacements de cellules qui contiennent des données 
nécessaires à la réalisation du calcul de ce résultat.

Une fonction est un calcul prédéfini saisi dans une cellule pour vous aider à analyser ou à manipu-
ler les données dans un classeur. Tout ce que vous avez à faire est d’ajouter les arguments, et le 
calcul sera automatiquement effectué pour vous. Les fonctions vous aident à créer les formules 
nécessaires pour obtenir les résultats que vous recherchez.

Paramétrer un classeur

Si vous avez l’intention d’utiliser plus qu’une simple feuille de calcul dans Calc, cela vaut la peine 
de faire quelques réglages pour éviter les pièges suivants :

• saisir des valeurs fixes dans les formules,

• oublier d’inclure des notes et des commentaires pour décrire ce que fait le système, y com-
pris les entrées nécessaires et l’origine des formules (si elles ne sont pas créées en partant 
de zéro).

• oublier d’incorporer un ensemble de tests pour vérifier que les formules font bien ce qui est 
prévu.

Le piège des valeurs fixes

De nombreux utilisateurs créent des formules longues et complexes en tapant directement des va-
leurs fixes dans celles-ci.

Par exemple, la conversion d’une monnaie dans une autre demande de connaître le taux courant 
de conversion. Si vous entrez dans la cellule C1 la formule =0,75*B1 (par exemple pour calculer 
la valeur en euros d’une somme en dollars américains qui est dans la cellule B1), vous devrez mo-
difier la formule quand le taux de change passera de 0,75 à une autre valeur. Il est bien plus 
simple de consacrer une cellule au taux de change et de faire référence à celle-ci dans toutes les 
formules qui utiliseront celui-ci. Les calculs de type « Et si » sont également simplifiés : que se 
passe-t-il si le taux de change varie de 0,75 à 0,70 ou 0,80 ? Aucune modification de la formule 
n’est nécessaire et on sait clairement quel est le taux utilisé dans les calculs. Partager une formule
complexe en éléments plus faciles à gérer, méthode décrite ci-dessous, aide aussi à minimiser les 
erreurs et facilite le dépannage.

Le manque de documentation

Le manque de documentation est une erreur très courante. Beaucoup d’utilisateurs commencent 
par une simple feuille de calcul qui se développe au cours du temps en quelque chose de beau-
coup plus compliqué. Sans documentation, le but initial et la méthodologie sont souvent peu clairs 
et difficiles à déchiffrer. Dans ce cas-là, il est souvent plus simple de recommencer depuis le début
en laissant de côté le travail fait jusque-là. Si vous insérez des commentaires dans les cellules et 

Paramétrer un classeur | 1



que vous utilisez des étiquettes et des titres, un classeur peut être modifié ultérieurement par vous 
ou par d’autres et beaucoup de temps et d’effort seront épargnés.

La vérification des formules

Additionner des colonnes de données ou des cellules sélectionnées dans une feuille de calcul pro-
duit souvent des erreurs dues à l’omission de certaines cellules, à une mauvaise définition de la 
plage ou à des cellules comptées deux fois. Il est utile d’instituer des tests dans vos classeurs. Par
exemple, vous créez un classeur où vous entrez des colonnes de nombres et où vous utilisez la 
fonction SOMME pour calculer les totaux de chaque colonne. Vous pouvez vérifier le résultat en 
ajoutant une colonne (non imprimée) contenant les totaux de chaque ligne et en calculant la 
somme de ceux-ci. Les deux résultats, total des lignes et total des colonnes, doivent être égaux, 
sinon il y a une erreur quelque part.

Il est même possible de définir une formule pour calculer la différence entre les deux totaux et affi-
cher un message d’erreur si elle n’est pas nulle (Figure 1).

Figure 1 : Vérification de formules.

Créer des formules

Vous pouvez saisir des formules de deux manières, soit en utilisant l’assistant Fonctions, soit en 
tapant directement dans la cellule elle-même ou dans la ligne de saisie. Dans les deux cas, une 
formule doit commencer par le symbole =, aussi, quand vous la saisissez directement, vous la de-
vez commencer par l’un des symboles suivants : =, + ou -. Calc ajoute automatiquement le sym-
bole = si vous commencez par le caractère + ou -. Si vous commencez par quelque chose d’autre, 
la formule sera considérée comme étant un texte simple.

Opérateurs dans les formules

Toutes les cellules d’une feuille peuvent être utilisées pour contenir des données ou le résultat de 
calculs. L’entrée de données s’effectue simplement en saisissant dans la cellule et en se déplaçant
vers la cellule suivante, ou en appuyant sur Entrée. Pour les formules, le signe égal indique que 
la cellule sera utilisée pour un calcul. Un calcul mathématique comme 15 + 46 peut être effectué 
comme indiqué Figure 2.

Si le calcul sur la gauche est effectué dans une seule cellule, la vraie puissance des formules est 
montrée sur la droite, où les données sont placées dans des cellules et le calcul est effectué en 
faisant référence à ces cellules. Dans ce cas, les cellules B3 et B4 contiennent les données, et B5 
est la cellule où le calcul est effectué. Notez que la formule s’affiche sous la forme =B3+B4. Le 
signe plus indique que les contenus des cellules B3 et B4 sont additionnés et donc que le résultat 
est inséré dans la cellule contenant la formule. Toutes les formules sont construites suivant ce prin-
cipe. D’autres façons de saisir des formules sont indiquées Tableau 1.

Ces références de cellules permettent aux formules d’utiliser les données où qu’elles soient dans 
la feuille en cours, ou dans d’autres feuilles du classeur qui est ouvert. Si les données nécessaires 
sont dans des feuilles différentes, elles seront référencées en utilisant le nom de la feuille, par 
exemple =SOMME(Feuille2.B12+Feuille3.A11).

Calcul simple dans une cellule Calcul par référence

Figure 2 : Un calcul simple.

Remarque

Pour saisir le symbole = dans un autre but que créer une formule comme décrit dans ce 
chapitre, tapez une apostrophe avant ce symbole. Par exemple, dans la saisie « '= peut 
signifier plusieurs choses pour des personnes différentes », Calc traite 
tout ce qui se trouve après l’apostrophe, et donc le signe =, comme du texte.

Tableau 1 : méthodes de saisie courantes de formules.

Formule Description

=A1+10 Affiche le contenu de la cellule A1 plus 10.

=A1*16% Affiche 16 % du contenu de la cellule A1.

=A1*A2 Affiche le résultat de la multiplication des cellules A1 et A2.

=ARRONDI(A1;1) Affiche le contenu de la cellule A1 arrondi à une décimale.

Formule Description

=TAUX.EFFECTIF(5%;12) Calcule le taux d’intérêt effectif pour un taux d’intérêt nominal
de 5 % et 12 paiements par an.

=B8-SOMME(B10:B14) Calcule B8 moins la somme des cellules de B10 à B14.

=SOMME(B8;SOMME(B10:B14)) Calcule la somme des cellules B10 à B14 et ajoute la valeur 
de B8.

=SOMME(B1:B1048576) Fait la somme de tous les nombres de la colonne B.

=MOYENNE(Glucide) Affiche la moyenne de la plage nommée Glucide.

=SI(C31>140; "HAUT"; "OK") Affiche le résultat de l’analyse conditionnelle de deux don-
nées. Si le contenu de C31 est supérieur à 140, alors HAUT 
est affiché, et sinon OK est affiché.

Les fonctions peuvent être identifiées dans le Tableau 1 par un mot, par exemple ARRONDI, suivi 
de parenthèses renfermant des références ou des nombres.

Remarque

Les utilisateurs d’autres tableurs sont peut-être habitués à des formules qui commencent 
par +, -, =, (, @, ., $ ou #. Les formules mathématiques peuvent ressembler à +D2+C2 ou à 
+2*3. Les fonctions commencent par le symbole @ comme @SUM(D2..D7), 
@COS(@DEGTORAD(30)) et @IRR(GUESS;CASHFLOWS). Les plages sont représentées 
comme A1..D3.

Il est également possible de définir des plages en les nommant, grâce à Insertion > Noms. Par 
exemple, Glucide représente une plage comme B3:B10. Des fonctions de logique peuvent aussi 
être mises en œuvre, comme l’expression SI qui aura comme résultat une réponse conditionnelle 
selon la donnée dans la cellule identifiée, par exemple :

=SI(A2>0;"Positif";"Négatif")

Une valeur de 3 dans la cellule A2 retournera le résultat Positif, -9 le résultat Négatif.

Types d’opérateurs

Vous pouvez utiliser les types d’opérateurs suivants dans LibreOffice Calc : arithmétiques, de com-
paraison, de texte et de référence.

Opérateurs arithmétiques

Les opérateurs d’addition, de soustraction, de multiplication et de division retournent des résultats 
numériques. Les opérateurs de négation et de pourcentage identifient une caractéristique du 
nombre qui se trouve dans la cellule, par exemple -37. L’exemple pour l’exponentiation illustre 
comment saisir un nombre qui est multiplié par lui-même un certain nombre de fois, par exemple 
2³ = 2 x 2 x 2.

Tableau 2. Opérateurs arithmétiques

Opérateur Nom Exemple

+ (plus) Addition =1+1

‒ (moins) Soustraction =2–1

‒ (moins) Négation –5

* (astérisque) Multiplication =2*2

/ (barre oblique) Division =10/5

% (pourcent) Pourcentage =15%

^ (accent circonflexe) Exponentiation =2^3

Opérateurs de comparaison

Les opérateurs de comparaison se rencontrent dans des formules qui utilisent la fonction SI et re-
tournent une réponse soit vraie soit fausse ; par exemple, =SI(B6>G12;127;0) signifie que si le 
contenu de la cellule B6 est supérieur au contenu de la cellule G12, Calc va retourner le nombre 
127, ou sinon Calc va retourner le nombre 0.

Une réponse directe VRAI ou FAUX peut être obtenue en saisissant une formule comme 
=B6>G12. Si les nombres trouvés dans les cellules référencées sont corrects et que celui de la 
cellule B6 est supérieur à celui de la cellule G12, la réponse VRAI est renvoyée, et dans le cas 
contraire la réponse FAUX est renvoyée.

Tableau 3. Opérateurs de comparaison

Opérateur Nom Exemple

= (signe égal) Égal A1=B1

> (supérieur à) Supérieur à A1>B1

< (inférieur à) Inférieur à A1<B1

>= (supérieur ou égal à) Supérieur ou égal à A1>=B1

<= (inférieur ou égal à) Inférieur ou égal à A1<=B1

<> (différence) Non égal A1<>B1

Si la cellule A1 contient la valeur numérique 4 et la cellule B1 la valeur numérique 5, les exemples 
ci-dessus retourneront respectivement les résultats FAUX, FAUX, VRAI, FAUX, VRAI et VRAI.

Opérateurs de texte

Les utilisateurs placent souvent des textes dans les classeurs. Pour amener de la flexibilité dans la
manière dont ce type de données sera affiché, les textes peuvent être assemblés à partir de frag-
ments provenant de différents endroits du classeur, comme montré Figure 3.

Dans cet exemple, des morceaux particuliers de texte se trouvent dans différentes cellules. Pour 
assembler ces segments, vous devez ajouter à la formule les espaces et ponctuations nécessaires
entre guillemets, d’où la formule résultante =B2 & ", "& B3 & " " & B5 & " " &B4. Le ré-
sultat est la concaténation en un texte représentant une adresse complète.

Figure 3 : Exemple de concaténation de texte.

Calc dispose d’une fonction CONCATENER qui réalise la même opération.

Opérateurs de référence

Une cellule individuelle est repérée par l’identifiant de sa colonne (lettre), situé en haut des co-
lonnes, et par l’identifiant de sa ligne (nombre), situé du côté gauche du classeur. Ainsi, dans les 
classeurs, la cellule supérieure gauche est A1.

Dans sa forme la plus simple, une référence se réfère à une seule cellule, mais elle peut aussi se 
référer à une plage rectangulaire ou à trois dimensions, ou encore à une autre référence dans une 
liste. Pour construire de telles références, vous avez besoin des opérateurs de référence.

Opérateur de plage

L’opérateur de plage est représenté par le signe deux-points (:). Une expression utilisant l’opéra-
teur de plage a la syntaxe suivante :

référence supérieure gauche:référence inférieure droite

L’opérateur de plage construit une référence à la plage comprise entre la référence gauche et la 
référence droite.

Figure 4 : Référence d’une plage de cellules sélectionnées

Dans le coin supérieur gauche de la Figure 4, la référence A1:D12 est affichée, ce qui correspond 
aux cellules incluses dans l’opération. Il est nécessaire de faire glisser la souris de A1 à D12 pour 
mettre cette plage en surbrillance.

Exemples :

A2:B4
Référence à une plage rectangulaire de 6 cellules, de 2 colonnes de large x 3 lignes de haut.
Quand vous cliquez sur cette référence dans une formule dans la ligne de saisie, une bor-
dure matérialise le rectangle.

(A2:B4):C9
Fait référence à une plage rectangulaire dont la cellule A2 occupe le sommet supérieur 
gauche et la cellule C9 le sommet inférieur droit. Cette plage contient donc 24 cellules, a une
largeur de 3 colonnes et une hauteur de 8 lignes. Cette méthode étend la plage A2:B4 initiale
à A2:C9.

Feuille1.A3:Feuille3.D4
Référence à une plage à trois dimensions de 24 cellules, de 4 colonnes de large sur 2 lignes 
de haut et 3 feuilles de profondeur.

B:B
Fait référence à toutes les cellules de la colonne B.

A:D
Fait référence à toutes les cellules des colonnes de A à D.

20:20
Fait référence à toutes les cellules de la ligne 20.

1:20
Fait référence à toutes les cellules des lignes de 1 à 20.

Si vous saisissez directement B4:A2 ou A4:B2, Calc le transformera alors en A2:B4. Ainsi, la cel-
lule en haut à gauche sera à gauche des deux-points, et la cellule en bas à droite sera à droite des
deux-points. Mais si vous nommez _debut la cellule B4  et _fin la cellule A2, vous pouvez utiliser 
_debut:_fin sans provoquer d’erreur.

Remarque

La liste déroulante Syntaxe de la formule de la page LibreOffice Calc – Formule de la boîte 
de dialogue Options (Outils > Options) permet d’utiliser les notations Excel A1 ou Excel 
L1C1. Il est toutefois recommandé de garder la syntaxe par défaut Calc A1.

Opérateur de concaténation de références

L’opérateur de concaténation de références est représenté par le signe tilde (~). Une expression 
utilisant l’opérateur de concaténation de références a la syntaxe suivante :

référence gauche~référence droite

Le résultat d’une telle expression est une référence de liste, qui est une liste ordonnée de réfé-
rences. Quelques fonctions acceptent une référence de liste en argument, comme SOMME, MAX 
ou INDEX par exemple.

La concaténation de références est parfois appelée « union ». Mais ce n’est pas l’union de deux 
ensembles « référence gauche » et « référence droite » telle qu’on l’entend dans la théorie des en-
sembles. NB(A1:C3~B2:D2) retourne la valeur 12 (9+3) si toutes les cellules contiennent un 
nombre, mais l’union des deux références contient uniquement 10 cellules.

Notez que SOMME(A1:C3;B2:D2) est différent de SOMME(A1:C3~B2:D2), bien que les résultats
soient les mêmes. La première expression est l’appel d’une fonction avec deux paramètres, cha-
cun étant une référence à une plage. La seconde expression est l’appel d’une fonction avec un pa-
ramètre, qui est une référence de liste.

La concaténation de références s’applique aussi à des lignes ou des colonnes entières. Par 
exemple SOMME(A:B~D:D) est la somme de toutes les cellules des colonnes A et B et de la co-
lonne D.

Opérateur d’intersection

L’opérateur d’intersection est représenté par le signe point d’exclamation (!). Une expression utili-
sant l’opérateur d’intersection a la syntaxe suivante :

référence gauche!référence droite

Si les références sont des plages simples, le résultat est une référence à une plage simple, qui 
contient toutes les cellules se trouvant à la fois dans la référence gauche et dans la référence 
droite.

Si les références sont des références de liste, alors chaque élément de la liste de gauche est croi-
sé avec chaque élément de la liste de droite, et ces résultats sont concaténés dans une référence 
de liste. L’ordre est tout d’abord de croiser le premier élément de la liste de gauche avec tous les 
éléments de la liste de droite, puis de croiser le deuxième élément de la liste de gauche avec tous 
les éléments de la liste de droite, et ainsi de suite.

Exemples :

A2:B4!B3:D6
Le résultat est une référence à la plage B3:B4, parce que ces cellules sont dans A2:B4 et 
dans B3:D6.

(A2:B4~B1:C2)!(B2:C6~C1:D3)
Les intersections A2:B4!B2:C6, A2:B4!C1:D3, B1:C2!B2:C6 et B1:C2!C1:D3 sont tout 
d’abord calculées. Le résultat est B2:B4, vide, B2:C2 et C1:C2. Puis ces résultats sont 
concaténés, en supprimant les parties vides. Le résultat final est donc la référence de liste 
B2:B4~B2:C2~C1:C2.

A:B!10:10
Calcule l’intersection des colonnes A et B avec la ligne 10, ce qui sélectionne donc A10 et 
B10.

Vous pouvez utiliser l’opérateur d’intersection pour faire référence à une cellule d’un tableau croisé
d’une façon explicite. Si vous avez des colonnes intitulées Températures et Précipitations, et des 
lignes intitulées Janvier, Février, Mars et ainsi de suite, alors l’expression suivante

'Février'!'Température'

fera référence à la cellule contenant la température du mois de février.

L’opérateur d’intersection (!) devrait avoir une priorité plus haute que l’opérateur de concaténation 
(~), mais ne vous fiez pas aux priorités.

Conseil

Mettez toujours entre parenthèses la partie qui doit être évaluée en premier.

Références relatives et absolues

Une référence, c’est-à-dire la façon dont on se réfère à l’emplacement d’une cellule particulière 
dans Calc, peut être soit relative (à la cellule en cours), soit absolue (fixe).

Référence relative

La Figure 5 est un exemple de référence relative qui illustre la différence entre les références rela-
tives et absolues dans les classeurs.

 1) À partir de l’exemple de la Figure 2, tapez les nombres 4 et 11 respectivement dans les cel-
lules C3 et C4.

 2) Copiez la formule de la cellule B5 dans la cellule C5. Vous pouvez le faire en utilisant un 
simple copier-coller ou en cliquant sur la poignée de B5 et faisant glisser vers C5, comme ci-
dessous. La formule de B5 calcule la somme des valeurs des cellules B3 et B4.

 3) Cliquez dans la cellule C5. La barre de formule affiche =C3+C4 au lieu de =B3+B4, et la va-
leur de C5 est 15, la somme de 4 et de 11 qui sont les valeurs dans C3 et C4.

Figure 5 : Références relatives.

Dans la cellule B5, les références aux cellules B3 et B4 sont des références relatives. Cela signifie 
que Calc interprète la formule de B5, l’applique aux cellules de la colonne B et dépose le résultat 
dans la cellule qui contient cette formule. Quand vous copiez la formule dans une autre cellule, la 
même procédure est appliquée pour calculer la valeur de cette nouvelle cellule. Cette fois, la for-
mule dans la cellule C5 se réfère aux cellules C3 et C4.

Vous pouvez comprendre une adresse relative comme étant une paire de déplacements depuis la 
cellule en cours. La cellule B1 est une colonne à gauche et quatre lignes au-dessus de la cellule 
C5. L’adresse pourrait être écrite L[-1]C[-4]. En fait, des tableurs plus anciens permettaient ce type 
de notation dans les formules.

À chaque fois que vous copiez cette formule de B5 dans une autre cellule, le résultat sera toujours 
la somme des deux nombres pris dans les deux cellules une ligne et deux lignes au-dessus de la 
cellule qui contiendra la formule.

L’adressage relatif est la méthode par défaut pour se référer à des adresses dans Calc.

Référence absolue

Vous voudrez peut-être multiplier une colonne de nombres par une valeur fixe, une colonne de 
montants en Dollars US par exemple. Pour convertir ces montants en Euros, il est nécessaire de 
multiplier chaque montant en dollars par le taux de change. 10,00 $ sera multiplié par 0,75 pour 
être converti en euros, ce qui donne ici 7,50 €. L’exemple suivant montre comment entrer un taux 
de change et l’utiliser pour convertir une colonne de montants de dollars en euros.

 1) Entrez le taux de change du dollar en euro (0,75) dans la cellule D1. Saisissez des montants
(en dollars) dans les cellules D2, D3 et D4, par exemple 10, 20 et 30.

 2) Dans la cellule E2, saisissez la formule =D2*D1. Le résultat est 7,5, ce qui est correct.

 
Saisissez la formule de conversion dans E2, le résultat est exact  ; copiez-le ensuite dans E3.

 3) Copiez la formule de la cellule E2 dans la cellule E3. Le résultat est 200, ce qui est manifes-
tement faux. Calc a copié la formule en utilisant l’adressage relatif. La formule dans E3 est 
=D3*D2 et non pas ce qui est voulu, soit =D3*D1.

 
Le résultat dans E3 est manifestement faux ; modifiez la formule dans E2 en tapant une référence

absolue.

 4) Dans la cellule E2, modifiez la formule en =D2*$D$1. Copiez-la dans les cellules E3 et E4. 
Les résultats sont alors 15 et 22,5, ce qui est correct.

 
Appliquez la formule correcte de E2 dans E3 pour obtenir le bon résultat.

Figure 6 : Références absolues.

Le signe $ avant le D et avant le 1 convertit la référence à la cellule D1 du type relatif au type ab-
solu ou fixe. Si la formule est copiée dans une autre cellule, la seconde partie sera toujours $D$1. 
L’interprétation de cette formule est « prendre la valeur de la cellule une colonne à gauche dans la 
même ligne et la multiplier par la valeur de la cellule D1 ».

Les références de cellules peuvent être effectuées de quatre façons.

Référence Explication

D1 Relative, depuis la cellule E3, c’est la cellule une colonne à gauche et deux lignes 
au-dessus

$D$1 Absolue, depuis la cellule E3, c’est la cellule D1

$D1 Partiellement absolue, depuis la cellule E3, c’est la cellule dans la colonne D et 
deux lignes au-dessus

D$1 Partiellement absolue, depuis la cellule E3, c’est la cellule une colonne à gauche et 
dans la ligne 1

Conseil

Pour changer de type de référence dans les formules, cliquez sur la référence de cellule et 
appuyez sur F4 pour basculer entre les quatre différents types de référence. Pour ne modi-
fier qu’une partie de la formule, sélectionnez les cellules dans la barre de formule et parcou-
rez les types de référence en appuyant sur F4.

La connaissance de l’usage des références relatives et absolues est essentielle si vous voulez co-
pier et coller des formules et lier des feuilles de calcul.

Plages nommées

Les cellules et plages de cellules peuvent recevoir un nom. Nommer les cellules et les plages 
améliore la lisibilité des formules et facilite la maintenance du document. Un exemple simple 
consisterait à nommer « Poids » la plage de cellules B1:B10 avant de faire la somme de tous les 
poids. La formule est =SOMME(B1:B10). Quand la plage B1:B10 a reçu le nom de Poids, vous 
pouvez transformer la formule en =SOMME(Poids). L’avantage en termes de lisibilité de la for-
mule est manifeste.

Un autre avantage est que toutes les formules qui utilisent un nom de plage comme argument sont
mises à jour quand cette plage change de position ou de taille. Par exemple, si le nom Poids dé-
signe maintenant la plage P10:P30, vous n’avez pas besoin d’adapter toutes les formules qui uti-
lisent Poids comme argument, il suffit de mettre à jour la plage nommée Poids avec le nouvel em-
placement et la nouvelle taille.

Pour attribuer un nom à une cellule ou à une plage, sélectionnez celle-ci puis utilisez Feuille > 
Plages ou expressions nommées > Définir dans la barre de menu. La boîte de dialogue ci-des-
sous (Figure 7) s’ouvre avec la plage prédéfinie. Il ne vous reste qu’à spécifier son nom et son 
étendue.

Figure 7 : La boîte de dialogue Définir un nom.

Vous pouvez aussi définir le nom d’une plage directement dans la feuille en la sélectionnant avant 
de saisir ce nom dans la boîte de saisie de nom à gauche de la barre de formule (Figure 8).

Figure 8 : Attribution d’un nom à une plage dans la boîte de nom.

Pour accéder rapidement à une plage nommée, sélectionnez son nom dans la liste déroulante de 
la boîte de nom. La plage apparaît à l’écran et est sélectionnée.

Expressions nommées

Vous pouvez aussi attribuer un nom à une formule longue et complexe. Pour cela, ouvrez la boîte 
de dialogue Définir un nom (Figure 7), saisissez la formule dans la boîte de saisie Plage ou ex-
pression de formule, entrez son nom et cliquez sur Ajouter.

Supposons, par exemple, que vous ayez à calculer dans les cellules C1 à C10 les longueurs d’un 
ensemble de cercles dont les rayons sont donnés dans les cellules B1 à B10. Définissez l’expres-
sion nommée PÉRIMÈTRE1 par =2*PI()*B1 puis cliquez sur Ajouter. Dans la cellule C1, tapez 
=PÉRIMÈTRE et appuyez sur Entrée. La formule est appliquée à la cellule C1. Copiez-la et col-
lez-la dans les autres cellules de C2 à C10 pour avoir la longueur de tous les cercles. Toutes les 
cellules de la plage C1:C10 contiennent l’expression =PÉRIMÈTRE.

1 Vous pouvez taper le nom en minuscules. Si vous le saisissez en majuscules, vous pourrez toujours
taper en minuscules dans la barre de formule, le nom passera automatiquement en majuscules si la 
formule ne comporte pas d’erreurs. Pour les lettres accentuées, le plus simple est de taper en 
minuscules dans une cellule, passer son contenu en majuscules (Format > Texte > Majuscules) 
puis de couper le contenu et le coller dans le champ Nom de la boîte de dialogue Définir un nom.

12 | Chapitre 7 – Formules et fonctions



Notez que les expressions nommées utilisent les mêmes règles d’adressage, absolu ou relatif, que
les formules.

Ordre de calcul

L’ordre de calcul se réfère à la séquence dans laquelle les opérations numériques sont effectuées. 
Les divisions et les multiplications sont effectuées avant les additions et les soustractions. Il existe 
une tendance naturelle à considérer que les calculs s’effectuent de gauche à droite, à l’instar d’une
équation qui serait lue comme un texte en français. En réalité, Calc évalue la formule en entier, 
puis, conformément aux règles de priorité, va morceler les formules pour exécuter les multiplica-
tions et les divisions avant les autres opérations. Par conséquent, vous devriez tester votre formule
pour vous assurer que le résultat correct est bien obtenu. Voici un exemple de l’ordre de calcul des
opérations :

Calcul de gauche à droite
1 + 3 * 2 + 3 = 11
1 + 3 = 4, puis 4 x 2 = 8, puis 8 + 3 = 11

Autre possibilité
1 + 3 * 2 + 3 = 20
1 + 3 = 4 puis 2 + 3 = 5 et 4 × 5 = 20

Calcul ordonné dans LibreOffice (conforme aux règles mathématiques)
1 + 3 * 2 + 3 = 10
Le programme effectue la multiplication 3 x 2 avant de s’occuper des nombres à ajouter.
3 x 2 = 6, puis 1 + 6 + 3 = 10.

Si vous souhaitez obtenir une des deux premières solutions, vous devez placer les opérations 
entre parenthèses :

((1 + 3) * 2 ) + 3 = (4 * 2) + 3 = 8 + 3 = 11 et

(1 + 3) * (2 + 3) = 4 * (2 + 3) = 4 * 5 = 20.

Remarque

Utilisez des parenthèses pour regrouper les opérations dans l’ordre que vous souhaitez ; par
exemple, =B4+G12*C4/M12 peut devenir =((B4+G12)*C4)/M12. L’utilisation des paren-
thèses permet aussi de rendre plus facilement compréhensibles les formules les plus com-
plexes.

Feuilles liées par des calculs

Une autre puissante fonctionnalité de Calc est la possibilité de lier des données de plusieurs 
feuilles. Nommer les feuilles avec pertinence peut être utile pour identifier quelles sont les données
employées. Un nom comme Salaires ou Ventes Sud, par exemple, est beaucoup plus significatif 
que Feuille1. La fonction FEUILLE() retourne le numéro d’ordre de la feuille dans une collection de
feuilles de classeur. Ce numéro d’ordre peut ne pas correspondre aux noms par défaut Feuille1, 
Feuille2, et ainsi de suite, si les feuilles sont ensuite déplacées à des endroits différents. À l’ouver-
ture d’un nouveau classeur, il n’a qu’une seule feuille de calcul.

Par exemple, si la formule =FEUILLE() est placée dans A1 sur la Feuille1, elle renvoie 1. Si vous 
déplacez la Feuille1 entre les feuilles 2 et 3, la valeur renvoyée par la fonction est changée en 2, 
car la Feuille1 est maintenant en deuxième position.

Pour illustrer comment obtenir des données d’autres feuilles, voici un exemple de tableau profes-
sionnel où une entreprise cumule les chiffres d’affaires et les coûts de ses magasins opérationnels 
sur une seule feuille de synthèse.

Feuille contenant les don-
nées pour le magasin 1

Feuille contenant les don-
nées pour le magasin 2

Feuille contenant les don-
nées pour le magasin 3

Feuille contenant les don-
nées cumulées de tous les 
magasins

Figure 9 : combiner les données de plusieurs feuilles dans une seule.

Les feuilles ont été construites avec des structures identiques. La manière la plus facile de le faire 
est la suivante ;

 1) Ouvrez un nouveau classeur. Élaborez la feuille du premier magasin, saisissez les données, 
formatez les cellules et préparez les formules pour les différentes sommes des lignes et des 
colonnes.

 2) Faites un clic droit sur l’onglet de la feuille et choisissez Renommer la feuille. Saisissez 
Magasin1.

 3) Faites à nouveau un clic droit sur l’onglet de la feuille et choisissez Déplacer/Copier la 
feuille (Figure 10). Dans la boîte de dialogue, cochez l’option Copier et sélectionnez - placer
en dernière position - dans le champ Insérer avant. Dans la zone Nouveau nom, remplacez 
Magasin1_2 par Magasin2. Répétez l’opération pour produire les feuilles Magasin3 et 
Cumul.

Figure 10 : Copier une feuille.

 4) Saisissez les données des magasins 2 et 3 dans leurs feuilles respectives. Chaque feuille 
est indépendante et rapporte les résultats des magasins individuels.

 5) Dans la feuille Cumul, cliquez dans la cellule K8. Tapez =, sélectionnez l’onglet Magasin1, 
cliquez dans la cellule K8, appuyez sur +, faites de même pour les feuilles Magasin2 et Ma-
gasin3 et appuyez sur Entrée. Vous obtenez alors dans la cellule K8 la formule 
=Magasin1.K8+Magasin2.K8+Magasin3.K8, qui additionne les ventes de plantes pour 
les trois magasins (dernière capture d’écran de la Figure 9).

 6) Copiez la formule, sélectionnez la plage K8:N18, choisissez Édition > Collage spécial > 
Collage spécial, décochez les cases Tout insérer et Formats dans la section Sélection de la 
boîte de dialogue et cliquez sur OK. Si un message d’avertissement apparaît, vous deman-
dant si vous voulez écraser les données existantes, cliquez sur Oui.

Figure 11 : avertissement en cas de collage dans une cellule non vide.

 7) Vous avez alors copié les formules dans chaque cellule tout en conservant le format que 
vous aviez paramétré dans la feuille de départ. Bien sûr, vous devrez remettre de l’ordre 
dans la feuille en supprimant les zéros des lignes non formatées.

Remarque

Le comportement par défaut de LibreOffice est de coller tous les attributs de la cellule d’ori-
gine (formats, notes, objets, chaînes de caractères et nombres).

L’Assistant fonction peut être également utilisé pour effectuer le lien. L’utilisation de cet assistant 
est décrite plus en détail dans la section sur les Fonctions.

Comprendre les fonctions

Calc comprend 350 fonctions pour vous aider à analyser et référencer les données. La plupart de 
ces fonctions sont destinées à être utilisées avec des nombres, mais un certain nombre d’autres 
sont utilisées avec des dates et heures, ou même des textes. Une fonction peut être aussi simple 
que l’addition de deux nombres, ou que le calcul de la moyenne d’une série de nombres. D’un 
autre côté, elle peut être aussi complexe que de calculer la variance d’un échantillon, ou la tan-
gente hyperbolique d’un nombre.

Le nom d’une fonction est ordinairement une description abrégée de ce que fait cette fonction. Par 
exemple, la fonction VA calcule la valeur actuelle d’un investissement, et BINHEX convertit un 
nombre binaire en un nombre hexadécimal. Les fonctions sont affichées entièrement en lettres 
majuscules, même si elles ont été saisies en minuscules. Calc les lit correctement si elles sont sai-
sies en minuscules ou avec une casse mélangée.

! Attention

Par défaut, LibreOffice utilise les noms de fonction en français (SOMME par exemple). Vous
pouvez travailler avec les noms en anglais (SUM par exemple) en sélectionnant Outils > 
Options > LibreOffice Calc > Formule et en cochant Utiliser les noms de fonction anglais. 
Le fait de cocher ou décocher cette option va modifier le nom des fonctions existantes dans 
votre classeur.

Le Guide Calc en français utilise les noms de fonctions en français.

Quelques fonctions de base sont à peu près similaires aux opérateurs. Exemples :

+ Cet opérateur additionne deux nombres pour produire un résultat. SOMME() a également la 
capacité d’additionner des cellules ou des groupes de plages contenant des nombres.

* Cet opérateur multiplie deux nombres pour produire un résultat. PRODUIT() est à la multipli-
cation ce que SOMME() est à l’addition.

Chaque fonction comporte un nombre d’arguments utilisés pour les calculs. Ces arguments 
peuvent avoir leur propre nom, ou pas. Votre rôle est d’entrer les arguments nécessaires pour exé-
cuter la fonction. Dans certains cas, les arguments ont des valeurs prédéfinies, et vous pouvez 
avoir besoin de vous référer à l’Aide de LibreOffice ou à l’Annexe B (Description des fonctions) de 
ce livre pour les comprendre. Le plus souvent cependant, un argument est une valeur que vous 
saisissez manuellement, ou une valeur déjà saisie dans une cellule ou une plage de cellules dans 
la feuille. Dans Calc, vous pouvez entrer les valeurs d’autres cellules en entrant leur référence ou 
leur nom, ou en les sélectionnant avec la souris, sauf dans certains cas. Si les valeurs dans ces 
cellules changent, le résultat de cette fonction est automatiquement mis à jour.

Compatibilité avec d’autres tableurs

Le jeu de fonctions de Calc implémente le standard ODFF (Open Document Formula Fonctions) 
publié par OASIS2 et par ISO/IEC 263003. Pour des raisons de compatibilité, les fonctions et leurs 
arguments ont, dans Calc, des noms pratiquement identiques à leurs contreparties dans Microsoft 
Excel. Cependant, tant Excel que Calc possèdent des fonctions que l’autre n’a pas. De temps en 
temps, des fonctions qui possèdent le même nom dans Calc et Excel ont des arguments différents 
ou des noms légèrement différents pour le même argument. Malgré tout, la majorité des fonctions 
peuvent être utilisées dans Calc comme dans Excel sans aucun changement.

Quand les arguments ou la valeur renvoyée par une fonction ne sont pas interopérablse avec 
Microsoft Excel, Calc place un suffixe dans le nom de cette fonction pour distinguer la fonction 
propre à Calc de celle d’Excel.

fonction_ADD et fonction_EXCEL2003

Les fonctions dont le nom se termine par _ADD ou _EXCEL2003 renvoient les mêmes résultats 
que les fonctions correspondantes de Microsoft Excel 2003 sans le suffixe. Utilisez les fonctions 
sans suffixe pour obtenir des résultats basés sur les standards internationaux.

fonction.XCL

Si le classeur est exporté vers Microsoft Excel, la fonction est exportée sous forme de la fonction 
MATH équivalente qui existe depuis Excel 2013. Si vous prévoyez d’utiliser le classeur avec une 
version plus ancienne d’Excel, utilisez soit la fonction PRECIS qui existe depuis Excel 2010, soit la
fonction XCL qui est exportée sous forme de fonction compatible avec toutes les versions d’Excel.

Comprendre la structure des fonctions

Toutes les fonctions ont une structure similaire. Si vous utilisez le bon outil pour saisir une fonction,
vous pouvez éviter d’apprendre sa structure, quoique ce soit toujours utile en cas de problème.

Pour donner un exemple type, la structure de la fonction qui compte toutes les cellules dont le 
contenu correspond aux critères de recherche est la suivante :

=BDNB(Base_de_données;Champ_de_base_de_données;Critères_de_recherche)

Parce qu’une fonction ne peut exister par elle-même, elle doit toujours faire partie d’une formule. 
Par conséquent, même si la formule n’utilise qu’une seule fonction, elle doit toujours commencer 
par un signe =. Quel que soit l’endroit dans la formule où la fonction se trouve, elle doit commen-
cer par son nom, comme BDNB dans l’exemple ci-dessus.

Après le nom de la fonction viennent ses arguments. Tous les arguments sont obligatoires, sauf 
ceux explicitement signalés comme optionnels. Les arguments sont ajoutés entre les parenthèses 
et sont séparés par des points-virgules, sans espace entre les arguments et les points-virgules.

Beaucoup d’arguments sont des nombres. Une fonction de Calc peut recevoir jusqu’à trente 
nombres comme arguments. Cela peut ne pas paraître beaucoup, mais il peut s’agir non seule-
ment un nombre ou une seule cellule, mais aussi un tableau ou une plage de cellules qui peut en 
contenir plusieurs, voire des centaines.

Selon la nature de la fonction, les arguments peuvent être saisis comme ceci :

2 https://docs.oasis-open.org/office/v1.2/os/OpenDocument-v1.2-os-part2.html
3 https://www.iso.org/standard/66375.html

"donnée texte" Les guillemets indiquent que du texte ou une chaîne de caractères est saisi.

9 Le nombre 9 est saisi en tant que nombre.

"9" Le nombre 9 est saisi en tant que texte.

A1 L’adresse de ce qui se trouve dans la cellule A1 est saisie.

Fonctions imbriquées

Les fonctions peuvent aussi être utilisées en tant qu’arguments à l’intérieur d’autres fonctions. 
Elles sont appelées des fonctions imbriquées.

=SOMME(2;PRODUIT(5;7))

Pour avoir une idée de ce que peuvent faire les fonctions imbriquées, imaginez que vous construi-
sez un module d’auto-apprentissage. Durant ce module, les étudiants remplissent trois question-
naires, dont les résultats sont mis dans les cellules A1, A2 et A3. En A4, vous pouvez créer une 
formule imbriquée qui commence par faire la moyenne des résultats de ce jeu de questions-ré-
ponses avec la formule =MOYENNE(A1:A3). La formule utilise alors la fonction SI pour donner aux
étudiants un retour correspondant à leur note moyenne. La formule en entier serait :

=SI(MOYENNE(A1:A3)>85;"Bravo ! Vous pouvez avancer au module 
suivant";"Echec. Veuillez revoir votre cours. Contactez votre enseignant 
si nécessaire.")

Selon sa moyenne, l’étudiant recevra un message de félicitation ou d’échec.

Notez qu’il ne faut pas inclure de signe = devant la fonction imbriquée (MOYENNE en l’occur-
rence). Le signe qui se trouve au début de la formule est suffisant.

Si l’utilisation des classeurs est nouvelle pour vous, le mieux est de penser aux fonctions comme à
un langage de script. Les exemples fournis sont simples pour expliquer les concepts plus claire-
ment, mais, à travers les fonctions imbriquées, une formule Calc peut vite devenir plus complexe.

Remarque

Calc vous suggère la syntaxe de la fonction dans une infobulle lors de la saisie.

Liste de fonctions

La Liste des fonctions (Figure 12) se trouve dans le volet latéral. Elle comprend une brève descrip-
tion de chaque fonction et de ses arguments ; mettez en surbrillance la fonction et regardez en bas
du panneau pour voir cette description. Si nécessaire, placez le curseur au-dessus de la sépara-
tion entre la liste et la description ; lorsque le curseur devient une double flèche avec lignes paral-
lèles, tirez la séparation vers le haut pour augmenter la place de la description. Faites un double-
clic sur le nom d’une fonction pour l’insérer dans la cellule en cours avec des substituants destinés
à être remplacés par les valeurs que vous aurez définies pour chacun de ses arguments.

La manipulation de la liste des fonctions est presque aussi rapide qu’une entrée manuelle, et a 
l’avantage de vous affranchir de la mémorisation d’une fonction que vous voulez utiliser. En théo-
rie, cette méthode devrait générer moins d’erreurs. En pratique cependant, certains utilisateurs 
peuvent commettre des maladresses en renseignant les valeurs dans les substituants. Une autre 
fonctionnalité est la possibilité d’afficher les dernières fonctions utilisées ou de les afficher par ca-
tégorie.

Figure 12 : La liste des fonctions dans le volet latéral.

Assistant Fonctions

L’Assistant Fonctions (Figure 13) est une méthode de saisie couramment utilisée. Pour ouvrir l’As-
sistant Fonctions, choisissez Insertion > Fonction dans la barre de menu, cliquez sur le bouton

 de la barre de formule, ou appuyez sur Ctrl+F2. Une fois ouvert, l’Assistant Fonctions a les 
mêmes fonctionnalités que la liste des fonctions, mais comporte en plus des champs dans les-
quels vous pouvez voir le résultat de la fonction finale, ainsi que le résultat d’une formule plus 
vaste dont elle fait partie.

Sélectionnez une catégorie de fonctions pour raccourcir la liste, puis faites-la défiler pour trouver le
nom de la fonction que vous souhaitez. Quand vous sélectionnez une fonction, sa description ap-
paraît du côté droit de la boîte de dialogue. Faites un double-clic sur la fonction souhaitée. Vous 
pouvez aussi commencer à taper le nom de la fonction dans le champ Rechercher. La liste des 
fonctions se réduit à chaque caractère que vous saisissez.

Figure 13 : La page Fonctions de l’Assistant Fonctions.

L’Assistant affiche alors une zone sur la droite où vous pouvez entrer manuellement des données 
dans les champs ou cliquer sur le bouton Sélectionner  pour réduire l’assistant, afin de pou-
voir sélectionner des cellules directement dans la feuille.

Figure 14 : L’Assistant Fonctions après réduction.

Pour sélectionner des cellules, cliquez directement dans la cellule ou gardez appuyé le bouton de 
la souris et faites glisser pour sélectionner la plage voulue.

Une fois la cellule ou la plage sélectionnée, cliquez sur le bouton Agrandir  pour retourner dans
l’Assistant.

Si plusieurs arguments sont nécessaires, sélectionnez le champ suivant et répétez autant de fois 
que nécessaire la procédure de sélection pour la cellule ou plage de cellules suivante. L’Assistant 
accepte jusqu’à 30 plages ou arguments pour la fonction SOMME.

Cliquez sur OK pour accepter la fonction, l’insérer dans la cellule et obtenir le résultat.

! Attention

Si vous sélectionnez une fonction par un double-clic dans la liste puis que vous changez 
d’avis et en sélectionnez une autre par un nouveau double-clic, la deuxième sera imbriquée 
dans la première dans la boîte de texte Formule. Vous devez vider celle-ci avant de faire un 
double-clic sur la deuxième fonction pour l’y ajouter.

Cette possibilité d’ajout de fonction imbriquée vous permet de construire des formules com-
plexes directement dans la boîte de texte Formule.

Vous pouvez également sélectionner l’onglet Structure pour voir l’arborescence des parties de la 
formule. Son principal avantage par rapport à la liste de fonctions est que chaque argument se 
trouve dans son propre champ, ce qui le rend plus facile à gérer. Le coût de cette fiabilité est une 
saisie plus lente, mais c’est souvent le prix à payer, étant donné que la précision est généralement 
plus importante que la vitesse pour créer une feuille.

La vue Structure de l’assistant est importante pour le débogage et la correction des formules très 
longues, très imbriquées et complexes. La formule y est décomposée et chacun de ses compo-
sants est calculé par un appel à une fonction plus simple ou à une opération arithmétique puis 
ceux-ci sont combinés en suivant les règles de calcul. Il est possible de visualiser chaque élément 
de la formule et de vérifier que les résultats intermédiaires sont exacts jusqu’à ce que l’erreur soit 
trouvée.

Les fonctions peuvent être entrées dans la ligne de saisie. Quand vous l’avez fait, appuyez sur la 
touche Entrée ou cliquez sur le bouton Accepter de la barre de formules pour insérer la fonction 
dans la cellule et obtenir son résultat.

1               2  3  4             5

1 Boîte de nom qui affiche la liste des fonc-
tions courantes

2 Assistant Fonctions 3 Annuler

4 Accepter 5 Ligne de saisie

Figure 15 : La barre de formules

Si la cellule affiche la formule à la place du résultat, l’option Formules est cochée dans la section 
Afficher de la page LibreOffice Calc – Affichage de la boîte de dialogue Options. Décochez-la 
pour afficher le résultat. Dans tous les cas, la formule reste visible dans la ligne de saisie.

Formules matricielles

Qu’est-ce qu’une formule matricielle ?

Une formule matricielle est une formule où sont calculées les valeurs d’une plage de cellules. À la 
différence des formules simples, les formules matricielles traitent plusieurs valeurs simultanément 
au lieu d’une seule. Une formule matricielle non seulement traite plusieurs valeurs, mais aussi ren-
voie plusieurs valeurs : le résultat d’une formule matricielle est aussi une matrice (tableau).

Quand Calc met à jour les formules, chaque cellule qui en contient une est lue et sa formule est re-
calculée. Si vous avez un millier de cellules dans une colonne qui contiennent la même formule 
(dont l’intitulé ne diffère que par la donnée utilisée pour le calcul), vous avez un millier de formules 
identiques à interpréter et à exécuter.

Une formule matricielle va évaluer la formule une seule fois et exécuter les calculs pour chaque 
élément du tableau ; le temps nécessaire pour évaluer la formule de chaque cellule est donc éco-
nomisé. Et puisque Calc ne stocke qu’une seule formule pour tout le tableau, il économise aussi 
de l’espace dans le fichier du classeur.

Figure 16 : La matrice source en jaune et la matrice résultat en vert.
La formule matricielle est visible dans la barre de formule.

Pour multiplier par 10 les valeurs de chaque cellule du tableau ci-dessus (Figure 16), il n’est pas 
nécessaire d’appliquer individuellement une formule à chacune de celles-ci. Il suffit juste d’utiliser 
une formule matricielle unique. Sélectionnez une plage de 3 cellules sur 5 à un autre endroit du 
classeur, saisissez la formule =10*A1:C5 et confirmez cette entrée par la combinaison de touches
Ctrl+Maj+Entrée. Le résultat est un tableau de 3 sur 5 dans lequel chaque cellule contient la 
valeur de la cellule correspondante de la plage (A1:C5) multipliée par 10.

En plus de la multiplication, vous pouvez utiliser d’autres opérateurs sur la plage de référence (la 
matrice). Avec Calc, vous pouvez ajouter (+), soustraire (‒), multiplier (*), diviser (/), élever à une 
puissance (^), concaténer (&) et comparer (=, <>, <, >, <=, >=). Ces opérateurs peuvent être utili-
sés sur chaque valeur de la plage de cellules et renvoient le résultat sous forme d’une matrice si 
une formule matricielle a été saisie.

Les opérateurs de comparaison traitent les cellules vides dans une formule matricielle de la même 
façon que dans une formule ordinaire, c’est-à-dire soit comme zéro, soit comme une chaîne vide. 
Par exemple, si les cellules A1 et A2 sont vides, les formules matricielles {=A1:A2=""} et 
{=A1:A2=0} renverront toutes les deux un tableau d’une colonne et deux lignes contenant la va-
leur VRAI.

Quand utiliser une formule matricielle ?

Utiliser des formules matricielles chaque fois que vous devez répéter des calculs avec des valeurs 
différentes. Si vous décidez plus tard de changer de méthode de calcul, vous n’aurez à modifier 
que la formule matricielle. Pour cela, sélectionnez le tableau tout entier puis apportez les modifica-
tions souhaitées à la formule matricielle.

Les matrices sont un outil essentiel quand il s’agit d’effectuer des calculs complexes, car plusieurs 
plages de cellules peuvent être incluses dans vos calculs. Calc possède plusieurs fonctions ma-
thématiques applicables aux matrices, comme la fonction PRODUITMAT qui multiplie deux ma-
trices.

Créer des formules matricielles

Si vous créez une formule matricielle avec l’Assistant Fonctions, vous devez cocher l’option Ma-
trice à chaque fois pour que les résultats soient renvoyés dans un tableau (en bas à gauche de la
Figure 13), sinon seule la valeur de la cellule en haut à gauche de celui-ci est renvoyée.

Si vous saisissez la formule matricielle directement dans la cellule, vous devez utiliser la combinai-
son de touche Ctrl+Maj+Entrée au lieu de la touche Entrée seule. La formule ne devient une 
formule matricielle qu’à cette condition.

Remarque

Les formules matricielles apparaissent dans Calc encadrées par des accolades, mais il n’est
pas possible de créer une formule matricielle en saisissant manuellement ces accolades.

! Attention

Les cellules de la matrice résultat sont automatiquement protégées contre les modifications. 
Vous pouvez cependant modifier ou copier la formule matricielle en sélectionnant toute la 
plage de cellules de cette matrice.

Stratégies pour créer des formules et des fonctions

Les formules qui font plus qu’un simple calcul ou une somme de valeurs de lignes ou de colonnes 
ont habituellement plusieurs arguments. Par exemple, l’équation classique de la cinématique

x=x0+v 0 t+
1
2
at2

calcule la position d’un corps connaissant sa position d’origine (x0), sa vitesse initiale (v0), son ac-
célération (a) et le temps passé pour se déplacer de sa position initiale à sa position finale (t).

Pour faciliter la présentation, c’est une bonne pratique de construire une feuille de manière iden-
tique à celle de la Figure 17. Dans cet exemple, les variables individuelles sont saisies dans des 
cellules de la feuille et aucune modification de formule (dans la cellule B9) n’est nécessaire.

Figure 17 : construction d’une formule avec des arguments

Vous pouvez adopter plusieurs approches générales quand vous créez une formule. Au moment 
de décider de la démarche à adopter, pensez aux autres personnes qui auront à utiliser le clas-

seur, à son évolution et aux changements qui pourraient être effectués dans l’utilisation de la for-
mule.

Si d’autres utilisateurs travaillent sur ce classeur, assurez-vous qu’il est facile de comprendre 
quelle saisie est nécessaire et à quel endroit. L’explication de l’objet du classeur, des bases des 
calculs, des saisies indispensables et des résultats obtenus est souvent placée dans la première 
feuille.

Un classeur que vous construisez aujourd’hui, avec nombre de formules compliquées, peut ne pas
être aussi évident à utiliser et à maîtriser 6 ou 12 mois plus tard. N’hésitez pas à user de commen-
taires et de notes afin de documenter votre travail.

Vous-même, en tant que concepteur du classeur, pouvez être conscient que vous ne devez pas 
utiliser des valeurs négatives ou nulles pour un argument particulier, mais si un autre utilisateur 
entre de telles valeurs, votre formule sera-t-elle suffisamment robuste, ou retournera-t-elle simple-
ment un message standard (souvent peu utile) ? C’est un bon usage de traquer les erreurs avec 
des instructions logiques ou à l’aide du formatage conditionnel.

Placer une formule unique dans chaque cellule

La stratégie la plus basique est de simplifier au maximum ce dont les formules ont besoin. Vous 
placez alors une formule unique dans chaque cellule appropriée. Cette stratégie ne peut être re-
commandée que pour les classeurs très simples ou « éphémères » (à usage unique).

Séparer une formule en plusieurs parties, puis les combiner

Cette deuxième stratégie ressemble à la première, mais vous allez séparer les longues formules 
en plusieurs parties plus petites, puis combiner ces parties en un tout. Beaucoup d’exemples de ce
type existent pour les calculs scientifiques et industriels, où les résultats intermédiaires sont sto-
ckés dans plusieurs endroits du classeur.

Dans tous les cas, vous devriez adopter les principes de base de création de formules décrits pré-
cédemment.

Accélérer les calculs

Les classeurs sont souvent utilisés pour traiter des données brutes et produire des résumés, des 
synthèses et des présentations d’informations utiles au décideur ou pour servir de source pour les 
rapports. Les données brutes peuvent provenir de mesures physiques, d’opérations commerciales 
ou même d’une collection de données de plusieurs natures. Les calculs appliqués à ces données 
brutes peuvent prendre du temps et durer plusieurs minutes, plusieurs heures, voire plusieurs 
jours. Des classeurs qui contiennent des milliers ou même des centaines de milliers de lignes et 
plusieurs colonnes sont courants dans les services financiers ou les laboratoires.

Insérer des formules dans chaque cellule, et effectuer des milliers d’interprétations et de calculs de
ces formules, est une erreur courante. Voici quelques recommandations pour accélérer les calculs.

Utiliser des formules matricielles pour les grandes quantités de données

Avec les formules matricielles, une seule formule est appliquée à l’ensemble des données. L’éco-
nomie de temps de calcul peut être importante avec de grandes quantités de données.

Utiliser des fonctions de consolidation

Les fonctions de consolidation effectuent des calculs sur des ensembles de données. SOMME, 
SOMME.SI, SOMMEPROD sont des exemples de fonctions de consolidations. Si vous avez, par 
exemple, une longue facture de matériaux, où les quantités doivent être multipliées par un prix uni-
taire puis additionnées pour produire un coût total, au lieu d’appliquer une formule à chaque ligne 
de la facture avant de les additionner, vous pouvez utiliser la formule 
SOMMEPROD(quantité;prix_initaire), où quantité et prix_unitaire sont des plages 
nommées qui représentent la facture. SOMMEPROD multiplie chaque cellule des données 
quantité par la cellule correspondante de prix_unitaire avant d’additionner tous les pro-
duits.

Une situation semblable se produit quand vous devez faire la somme d’un sous-ensemble des 
données originales en appliquant un test à chaque entrée pour savoir s’il faut l’inclure dans la 
somme, par exemple si une valeur est strictement positive. Utilisez 
SOMME.SI(données_à_tester;">0";données_à_additionner), où données_à_tester 
est l’ensemble des données où vous recherchez les valeurs positives, données_à_additionner
est la colonne où se trouvent les valeurs à ajouter d’après le test et ">0" est le test en question.

Entre autres fonctions de consolidation, on trouve SOMME.SI.ENS, MOYENNE.SI, NB.SI, MIN.SI, 
MAX.SI et d’autres encore.

Utiliser l’éditeur Basic pour créer des fonctions

Une troisième stratégie est d’utiliser l’éditeur Basic et de créer vos propres fonctions et macros. 
Cette approche peut être utilisée pour simplifier l’utilisation du classeur par l’utilisateur final et gar-
der des formules simples avec un risque d’erreur limité. Cette approche peut également rendre la 
maintenance plus facile en centralisant les modifications et mises à jour en un seul endroit. L’utili-
sation des macros décrite dans le chapitre 12 est un sujet à part entière. Le danger de la sur-utili-
sation des macros et des fonctions personnalisées réside dans le fait que les principes sur les-
quels le classeur se base peuvent devenir plus difficiles à appréhender par un utilisateur autre que 
l’auteur d’origine (et parfois même par l’auteur !).

Détecter et corriger les erreurs

Des situations où des erreurs sont affichées se produisent souvent. Même avec tous les outils dis-
ponibles dans Calc pour vous aider à entrer des formules, commettre des erreurs est fréquent. 
Beaucoup de personnes trouvent la saisie de nombres difficile et peuvent commettre des erreurs 
sur le type d’entrée qu’un argument d’une fonction nécessite. En plus de corriger les erreurs, vous 
pouvez vouloir trouver les cellules utilisées dans une formule pour changer leurs valeurs ou vérifier
la réponse.

Calc dispose de trois outils pour réaliser des investigations dans les formules et les cellules 
qu’elles référencent : les messages d’erreur, le codage en couleur et l’Audit.

Messages d’erreur

Le plus simple pour gérer les erreurs est d’utiliser les messages d’erreur. Les messages d’erreurs 
s’affichent à la place du résultat dans la cellule d’une formule ou dans l’Assistant Fonctions.

Un message d’erreur pour une formule est habituellement un nombre à trois chiffres de 501 à 532, 
ou parfois un bout de texte comme #NOM?, #REF ! ou #VALEUR !. Le numéro de l’erreur apparaît
généralement dans la cellule et une courte explication de l’erreur est affichée dans la barre d’état.

La plupart des messages d’erreur indiquent un problème avec la façon dont la formule est saisie, 
mais d’autres peuvent indiquer que vous avez dépassé une limite de Calc ou de son paramétrage 
en cours.

Les messages d’erreurs ne sont pas toujours conviviaux et peuvent intimider les nouveaux utilisa-
teurs. Toutefois, ce sont des indices utiles pour corriger les erreurs. Vous pouvez trouver des expli-
cations détaillées pour chacun d’entre eux dans l’Aide, en recherchant dans l’index Message d’er-
reur ###, ou dans l’Annexe C. Quelques messages parmi les plus fréquents sont expliqués dans la
table suivante.

###### Il ne s’agit pas d’une erreur, mais simplement d’une colonne qui n’est pas assez
large pour afficher le résultat.

#NOM ? (525) L’identificateur (de nom, d’étiquette de ligne ou de colonne, de macro…) est in-
connu.

#REF ! (524) La ligne, colonne ou feuille référencée n’existe pas.

#VALEUR ! (519) La valeur de l’un des arguments n’est pas du type correct. Par exemple, des 
guillemets manquent autour d’une valeur, ou une cellule référencée contient un 
texte au lieu d’un nombre.

509 Un opérateur manque dans la formule.

510 Un argument manque dans la formule.

#DIV/0 ! (532) Division par zéro.

#NUM ! Le résultat d’un calcul se trouve hors de la plage de définition de la valeur.

Exemples d’erreurs fréquentes

#DIV/0 ! Division par zéro

Figure 18 : Exemples d’erreurs #DIV/0 ! Division par zéro.

L’erreur résulte de la division d’un nombre par le nombre zéro (0) ou par une cellule vide. Vous 
pouvez éviter facilement ce type de problème. Utilisez une fonction conditionnelle testant une va-
leur zéro ou une cellule vide. La Figure 18 montre la division de la colonne B par la colonne C, 
avec l’apparition de deux erreurs dues à un zéro ou une cellule vide dans la colonne C.

Il est très fréquent de rencontrer de telles erreurs dans une situation où des données sont man-
quantes ou ne sont pas saisies correctement. Quand une telle possibilité existe, une fonction SI 
peut être utilisée pour afficher les données correctement. La formule =SI(C3>0;B3/C3;"Pas de
valeur") peut être saisie. Cette formule peut alors être copiée dans le reste de la colonne D. La 
signification littérale de cette formule serait : Si C3 est supérieur à 0, alors diviser B3 par C3, sinon 
afficher "Pas de valeur" (Figure 19).

Il est également possible, dans le dernier argument, d’utiliser un nombre standard à utiliser par dé-
faut (0 par exemple). Une pratique courante est d’utiliser dans ce cas, des doubles guillemets ("") 
pour afficher un texte vide. Attention de ne recourir à cette solution que pour des cellules ne devant
pas être considérées comme des nombres dans des formules, sous peine de déclencher une er-
reur #VALEUR ! (voir ci-dessous).

Figure 19 : Solution pour la division par zéro

#VALEUR ! Valeur incorrecte et #REF ! Référence incorrecte

Figure 20 : Saisie incorrecte provoquant l’erreur #VALEUR !

L’erreur #VALEUR ! est également très fréquente, par exemple quand une cellule contient une 
donnée d’un type incorrect. Dans l’exemple de la Figure 20, le texte "Inconnu" a été saisi dans la 
cellule C8, alors que la formule de la colonne D attend à cet endroit un nombre.

L’erreur #REF ! est provoquée par une référence manquante. Dans l’exemple de la Figure 21, la 
formule fait référence à une feuille qui a été supprimée.

Figure 21 : une feuille supprimée provoque l’erreur #REF !

Remarque

Pour éviter les erreurs #VALEUR ! et #REF !, il est souvent utile de donner à une cellule, 
comme B17 dans la Figure 23, un nom comme TotalNord. Ainsi, le programme va transpor-
ter ce nom durant la copie de la formule, et cela évite d’employer une référence absolue 
($B$17).

Codage en couleur

Un autre outil utile pour la révision des formules est le codage en couleur. Quand vous éditez une 
formule déjà saisie, les cellules ou les plages utilisées dans chaque argument ont un contour en 
couleur.

Calc utilise huit couleurs pour entourer les cellules référencées, en commençant par le bleu pour la
première cellule, puis en continuant par le rouge, le magenta, le vert, le bleu marine, le marron, le 
violet et le jaune, et en répétant cette séquence autant de fois que nécessaire.

Figure 22 : Codage en couleur

Mise en évidence des valeurs

Dans certaines situations, le contenu d’une cellule peut s’afficher de façon identique avec des 
types de données différents. Par exemple, un contenu textuel ou numérique peut avoir le même 
aspect mais produire une erreur s’ils sont tous les deux utilisés dans des calculs. Par exemple, la 
chaîne "10,35" alignée à droite dans la cellule peut être confondue avec la valeur 10,35. Quand
cette cellule est utilisée dans une formule, la chaîne prend la valeur zéro qui peut produire une er-
reur.

La mise en évidence des valeurs fera la distinction entre les types texte et valeur numérique en af-
fectant des couleurs différentes aux caractères affichés. Par défaut, le texte est en caractères noirs
et les nombres en bleu. Voyez le chapitre 2 pour plus d’informations sur la mise en évidence des 
valeurs.

L’Audit

Dans les classeurs longs ou complexes, le codage en couleur devient moins utile. Dans ce cas, 
vous pouvez utiliser le sous-menu dans Outils > Audit. L’Audit est un outil pour vérifier quelles 
sont les cellules utilisées dans les arguments d’une formule (antécédents), ou dans quelles autres 
formules elle est utilisée (dépendants). Il peut également être utilisé pour repérer les erreurs, mar-
quer les données incorrectes (les informations dans les cellules qui n’ont pas le format correct pour
l’argument d’une fonction), ou supprimer les repères des antécédents et des dépendants.

Pour utiliser l’Audit, sélectionnez une cellule contenant une formule, puis démarrez l’Audit. Dans le 
classeur, vous verrez des lignes se terminant par des cercles pour indiquer les antécédents et des 
lignes se terminant par des flèches pour les dépendants. Ces lignes montrent le flux des informa-
tions. Faites un double-clic sur le repère pour sélectionner la cellule référencée.

Utilisez l’Audit pour vous aider à suivre les antécédents référencés par une formule dans une cel-
lule. En suivant ces antécédents, vous pourrez souvent trouver la source des erreurs. Placez le 
curseur dans la cellule en question, puis choisissez Outils > Audit > Repérer les antécédents 
dans la barre de menus ou appuyez sur Maj+F7. La Figure 23 montre un exemple simple de repé-
rage des antécédents.

Figure 23 : Audit > Repérer les antécédents – Résultat

Vous vous interrogez sur le nombre affiché dans la cellule B17, qui vous paraît erroné. La cause 
peut être détectée dans les cellules marquées d’un cercle bleu. Dans ce cas, la cellule B5 contient 
par erreur un signe négatif. Supprimer ce signe résout tout simplement le problème.

Dans d’autres cas, vous devrez tracer l’erreur. Utilisez la fonction Outils > Audit > Repérer les er-
reurs pour repérer les cellules à l’origine de l’erreur.

Exemples de fonctions

Pour les novices, les fonctions sont l’une des fonctionnalités les plus intimidantes de LibreOffice 
Calc. Les nouveaux utilisateurs apprennent rapidement que les fonctions sont des ressources im-
portantes des classeurs. On peut en dénombrer plus de 400, et beaucoup d’entre elles nécessitent
des paramètres qui reposent sur un savoir spécialisé. Heureusement, Calc comprend un certain 
nombre de fonctions à la portée de tous.

Fonctions de base arithmétiques et statistiques

Les fonctions les plus simples créent des formules d’arithmétique de base ou pour calculer des 
nombres dans une plage de cellules.

Arithmétique de base

Les fonctions de base de l’arithmétique sont l’addition, la soustraction, la multiplication et la divi-
sion.

Certaines de ces opérations disposent de leur propre fonction :

• SOMME pour l’addition ;

• PRODUIT pour la multiplication ;

• QUOTIENT pour la division.

Traditionnellement, aucune fonction n’est associée à la soustraction.

SOMME, PRODUIT et QUOTIENT sont utiles quand on veut saisir des plages de cellules de la 
même façon que pour les autres fonctions, avec les arguments entre parenthèses après le nom de
la fonction.

Cependant, pour les opérations simples, beaucoup d’utilisateurs préfèrent les symboles tradition-
nels, en utilisant le signe plus (+) pour l’addition, le signe moins (‒) pour la soustraction, l’asté-
risque (*) pour la multiplication et la barre oblique (/) pour la division. Ces symboles sont plus ra-
pides à saisir.

Un choix identique est disponible si vous voulez élever un nombre à la puissance d’un autre. Au 
lieu de saisir =PUISSANCE(A1;2), vous pouvez saisir =A1^2.

De plus, les symboles ont l’avantage de vous faire saisir les formules dans l’ordre habituel du rai-
sonnement humain, plutôt que l’ordre du traitement informatique employé par sa fonction équiva-
lente. Par exemple, au lieu de saisir =SOMME(A1:A2), ou =SOMME(A1;A2), vous saisissez 
=A1+A2 (voir l’astuce suivante). Cet ordre est souvent préférable pour les opérations composées, 
quand saisir =A1*(A2+A3) est plus court et plus facile à lire que 
=PRODUIT(A1;SOMME(A2:A3)).

Le principal inconvénient de l’utilisation des opérateurs arithmétiques réside dans le fait que vous 
ne pouvez pas utiliser directement une plage de cellules. En d’autres termes, pour saisir l’équi-
valent de =SOMME(A1:A4), vous devrez saisir =A1+A2+A3+A4.

Un autre inconvénient est que si l’on insère par exemple une cellule ou une ligne entre A1 et A4, 
une formule utilisant les opérateurs n’inclura pas la nouvelle cellule contrairement à une formule 
utilisant une plage.

Sinon, le fait d’utiliser une fonction ou un opérateur est surtout une question de choix personnel, 
sauf, évidemment, pour la soustraction. Cependant, si vous utilisez régulièrement des classeurs 
dans un groupe de travail, comme une classe ou un bureau, vous souhaiterez peut-être définir un 
standard d’utilisation de l’un ou de l’autre pour que tous ceux qui gèrent un classeur prennent l’ha-
bitude de ce standard.

Astuce

La fonction SOMME étant très utilisée, la barre de formule comprend un bouton qui insère 
cette fonction dans la cellule courante. LibreOffice suggère automatiquement une plage de 
cellules, au-dessus ou à gauche. Si la plage contient des filtres, la fonction SOUS.TOTAL 
est insérée à la place de la fonction SOMME.

Statistiques de base

Une autre utilisation fréquente des fonctions dans un classeur consiste à obtenir des informations 
d’une liste, comme la série de notes d’une classe ou les ventes d’une société par produit et par tri-
mestre.

Vous pouvez bien sûr parcourir une liste de nombres si vous voulez des informations simples, 
comme le nombre le plus haut, le plus bas ou la moyenne, par exemple. Le souci est que, plus la 
liste s’allonge, plus vous allez y passer du temps et plus le risque d’erreur augmente. Il est donc 
plus rapide et plus efficace d’utiliser une fonction. Ceci explique l’existence d’une fonction comme 
NB, qui ne fait que compter combien de nombres sont présents dans la plage de cellules spécifiée.

De même, pour trouver le nombre le plus haut ou le plus bas, vous pouvez utiliser MIN ou MAX. 
Pour chacune de ces fonctions, tous les arguments peuvent être soit une plage de cellules, soit 
une série de cellules entrées individuellement.

Chacune a une fonction proche, MINA ou MAXA, qui fait la même chose, mais considère que la 
valeur des cellules formatées en texte est 0 (d’autres fonctions avec un "A" supplémentaire à la fin 
existent). Cela peut être utile si vous utilisez une note de texte pour indiquer, par exemple, qu’un 
élève était absent à un examen et que vous voulez vérifier si un examen complémentaire est né-
cessaire.

Pour plus de souplesse pour des opérations similaires, vous pouvez utiliser les fonctions 
GRANDE.VALEUR ou PETITE.VALEUR, qui comportent toutes les deux un argument supplémen-
taire pour le rang. Si le rang utilisé avec GRANDE.VALEUR est 1, vous obtiendrez le même résul-
tat qu’avec MAX. Si le rang est 2, le résultat sera alors le deuxième résultat le plus grand. De 
même, si le rang est 2, PETITE.VALEUR vous donnera le deuxième nombre le plus petit. 
GRANDE.VALEUR et PETITE.VALEUR seront utiles pour un contrôle permanent et, en changeant
l’argument rang, vous pourrez rapidement obtenir plusieurs résultats.

Les experts trouveront dans Calc des fonctions spécialisées comme la loi de Poisson, la loi nor-
male ou la loi normale inverse d’un échantillon. Pour les autres, il y a des fonctions statistiques 
plus simples que vous pouvez rapidement apprendre à utiliser.

En particulier, si vous avez besoin d’une moyenne, vous avez le choix entre plusieurs fonctions. 
Vous pouvez calculer la moyenne arithmétique, c’est-à-dire la somme des valeurs d’une liste divi-
sée par le nombre de ces valeurs, en utilisant MOYENNE, ou MOYENNEA pour inclure les valeurs
texte et leur donner une valeur de zéro.

Vous pouvez de plus obtenir d’autres informations de l’ensemble de données :

• MEDIANE : la valeur de la liste qui se trouve exactement à mi-chemin entre le nombre le 
plus grand et le nombre le plus petit.

• MODE : la valeur la plus fréquente dans la liste.

• QUARTILE : la valeur à une position déterminée dans un tableau de nombres. Après la plage
de cellules, vous devez saisir le type de quartile : 0 pour la valeur minimale (équivalent à 
MIN), 1 pour la valeur à 25 %, 2 pour la valeur à 50 %, 3 pour la valeur à 75 % et 4 pour la 
valeur maximale (équivalent à MAX). Notez que le résultat pour les types de 1 à 3 peut ne 
pas correspondre à une valeur de la liste.

• RANG : la position d’une valeur de la liste donnée dans la liste entière, en suivant les valeurs
croissantes ou décroissantes. Vous devez saisir l’adresse de la cellule dont vous cherchez le
rang, la plage de cellules et le type de rang (ordre décroissant si le type est 0 ou vide, ordre 
croissant sinon).

Certaines de ces fonctions se recouvrent ; par exemple, MIN et MAX peuvent également être obte-
nues avec QUARTILE. Dans d’autres cas, un tri ou un filtre personnalisés peuvent donner le 
même résultat. Votre choix dépendra de vos goûts et de vos besoins. Certains préféreront utiliser 
MIN et MAX parce que c’est plus facile à retenir, tandis que d’autres préféreront QUARTILE parce 
que c’est plus puissant.

Utiliser ces fonctions

Dans certains cas, vous pouvez obtenir des résultats similaires à certaines de ces fonctions en pa-
ramétrant un filtre ou un tri personnalisés. Cependant, en général, les fonctions se révèlent être 
plus facilement ajustables que les filtres ou les tris et comportent une plus grande variété de possi-
bilités.

Certaines fois, vous aurez peut-être envie de saisir temporairement une ou plusieurs formules 
dans une cellule vide disponible, et de les supprimer lorsque vous avez terminé. Cependant, si 
vous vous apercevez que vous utilisez toujours les mêmes formules, vous pouvez songer à créer 
un modèle en laissant de l’espace pour toutes les formules que vous utilisez et en utilisant la cel-
lule à leur gauche pour saisir une étiquette. Une fois le modèle créé, vous pouvez facilement 
mettre à jour chaque formule après avoir saisi des valeurs, soit automatiquement, soit « à la vo-
lée » en appuyant sur la touche F9 pour actualiser toutes les cellules sélectionnées.

Quelle que soit votre façon de les utiliser, ces fonctions vous paraîtront probablement simples à 
utiliser et adaptables à de nombreux besoins. Quand vous aurez maîtrisé celles que nous venons 
de voir, vous serez prêt à en essayer de plus complexes.

Arrondir les nombres

Pour les besoins des mathématiques et des statistiques, Calc dispose de plusieurs façons d’arron-
dir les nombres. Si vous êtes programmeur, vous connaissez peut-être déjà certaines de ces mé-
thodes. Toutefois, vous n’avez pas besoin d’être un spécialiste pour apprécier leur utilité. Vous 
voudrez peut-être arrondir des valeurs monétaires, ou limiter le nombre de chiffres décimaux qui 
ne représentent rien dans le monde réel. Par exemple, si les articles que vous commandez sont 
conditionnés par 100, le fait que vous n’en ayez besoin que de 66 au maximum est sans intérêt 
pour vous ; vous devrez arrondir ce nombre pour pouvoir commander. En connaissant les options 
d’arrondi, vous pourrez rendre vos classeurs immédiatement plus opérationnels.

! Attention

Vous pouvez utiliser l’option Outils > Options > LibreOffice Calc > Calcul > Limiter les 
décimales pour le format de nombre général ou le champ Décimales de l’option Format 
> Cellules > Nombres pour arrondir un nombre. Cependant, soyez bien conscient que l’ar-
rondi ne s’applique qu’à l’affichage de la cellule, et que son contenu reste inchangé. Par 
exemple, si la valeur de la cellule A1 est de 1,2345 et que vous la formatez avec deux déci-
males, 1,23 s’affichera dans A1. Toutefois, si, dans la cellule A2 avec le format Standard, 
vous saisissez la formule =A1, c’est bien 1,2345 qui s’affichera.

Si vous utilisez une fonction d’arrondi, vous avez deux possibilités pour construire vos formules. 
Vous pouvez imbriquer un calcul dans une fonction d’arrondi. Par exemple, la formule 
=ARRONDI(SOMME(A1:A2)) additionne les nombres des cellules A1 et A2, puis arrondit le résul-
tat à l’entier le plus proche. Cependant, même si vous n’avez pas besoin du résultat exact tous les 
jours, vous pouvez vouloir vous y référer occasionnellement. Si c’est le cas, vous feriez alors 
mieux de séparer les deux fonctions, en plaçant =SOMME(A1:A2) dans la cellule A3, 
=ARRONDI(A3) dans A4, et en mettant une étiquette explicite pour chacune des cellules.

Voir l’aide pour plus de détails sur les méthodes d’arrondi.

Utiliser les caractères génériques et les expressions régulières
dans les fonctions

En informatique, un caractère générique est un caractère qui peut être remplacé par zéro ou plu-
sieurs caractères dans une chaîne. Ils sont d’un usage courant en programmation, dans les re-
quêtes des bases de données et pour naviguer dans les dossiers quand on utilise l’invite de com-
mande. Les caractères couramment employés sont l’astérisque (*) qui représente un caractère ou 
plus et le point d’interrogation (?) qui n’en représente qu’un seul.

Une expression régulière est une suite de caractères qui définit un motif de caractères recherchés.
Celui-ci est habituellement utilisé par les algorithmes des opérations « Rechercher » et « Recher-
cher & remplacer » sur les chaînes. Les expressions régulières sont plus puissantes que les carac-
tères génériques quand il s’agit d’extraire ou de remplacer des éléments de texte.

Calc utilise les caractères génériques et les expressions régulières pour la recherche dans les ar-
guments textuels de beaucoup de ses fonctions. Les expressions régulières sont définies par le 
projet ICU4

Remarque

Les expressions régulières et les caractères génériques sont mutuellement exclusifs : un 
seul des deux peut avoir la valeur VRAI. En interne, si les deux sont vrais, les caractères gé-
nériques ont la priorité sur les expressions régulières.

Certaines fonctions de Calc autorisent l’utilisation de caractères génériques : REGEX, SOMME.SI, 
NB.SI, EQUIV, CHERCHE, RECHERCHE, RECHERCHEH, RECHERCHEV, BDNB, BDNBVAL, 
BDSOMME, BDPRODUIT, BDMAX, BDMIN, BDMOYENNE, BDECARTYPE, BDECARTYPEP, 
BDVAR, BDVARP, BDLIRE.

4 http://userguide.icu-project.org/strings/regexp

Figure 24 : Paramètres par défaut de la page Calcul. Les caractères génériques et les expres-
sions régulières sont mutuellement exclusifs.

Les caractères génériques peuvent être employés si l’option Autoriser les caractères génériques 
dans les formules est cochée, dans l’option Outils > Options > LibreOffice Calc > Calcul (Figure
24).

Par exemple, si Autoriser les expressions régulières dans les formules est cochée, 
=NB.SI(A1:A6;"r.d") va prendre en compte les cellules dans A1:A6 qui contiennent rad et 
ROD.

De plus, si l’option Critères de recherche = et <> doivent correspondre à des cellules entières n’est
pas cochée, les cellules Fred, bride et Rideau seront également prises en compte. Si ce paramètre
est coché, il peut être contourné en modifiant l’expression : =NB.SI(A1:A6;".*r.d.*").

Les recherches par expressions régulières à l’intérieur des fonctions sont toujours insensibles à la 
casse, que l’option Respecter la casse de la page Outils > Options > LibreOffice Calc > Calcul 
soit cochée ou non. rad et ROD seront toujours pris en compte comme dans l’exemple de la Figure
25. Cette insensibilité à la casse s’applique également aux structures d’expressions régulières 
([:lower:]) et ([:upper:]).

Figure 25 : utilisation de la fonction NB.SI avec une expression régulière.

Les expressions régulières ne fonctionnent pas comme de simples comparaisons. Par exemple, 
A1="r.d" renverra FAUX si A1 contient rad, même si les caractères génériques sont autorisés. Il ne 
retournera VRAI que si A1 contient r.d (r suivi d’un point suivi de d). SI vous voulez tester des ex-

Utiliser les caractères génériques et les expressions régulières dans les fonctions | 35



pressions avec des expressions régulières, essayez la fonction NB.SI : NB.SI(A1;"r.d") re-
tournera 1 ou 0, interprété comme VRAI ou FAUX dans des formules telles que 
=SI(NB.SI(A1;"r.d");"Oui";"Non").

Si vous activez l’option Autoriser les expressions régulières dans les formules, cela signifie que 
pour toutes les fonctions ci-dessus, si vous avez besoin des caractères spéciaux des expressions 
régulières (comme les parenthèses) en tant que caractères ordinaires d’une chaîne dans une for-
mule, vous devrez les faire précéder par une barre oblique inversée (\), même s’ils ne font pas par-
tie d’une expression régulière. Ces barres obliques inversées devront être supprimées si, par la 
suite, cette l’option est désactivée.

Fonctions avancées

Comme c’est le cas avec d’autres tableurs, LibreOffice Calc peut être amélioré par des fonctions 
définies par les utilisateurs ou des modules externes. La construction de fonctions personnalisées 
peut être effectuée avec l’éditeur Basic EDI ou en écrivant des modules externes ou des exten-
sions séparées.

Les bases de l’écriture et de l’exécution des macros sont traitées chapitre 12, Macros Calc. Les 
macros peuvent être liées à des menus ou à des barres d’outils pour faciliter leur utilisation, ou 
être stockées dans des modules standard afin que les fonctions puissent être disponibles dans 
d’autres documents.

Les modules externes de Calc sont des extensions bureautiques spécialisées qui peuvent étendre 
les fonctionnalités de LibreOffice avec de nouvelles fonctions intégrées à Calc. L’écriture des mo-
dules externes nécessite la connaissance du langage C++ et du SDK de LibreOffice, et est réser-
vée aux programmeurs expérimentés. Un certain nombre d’extensions pour Calc ont été écrites et 
peuvent être trouvées sur le site des Extensions de LibreOffice. Reportez-vous au chapitre 14, Pa-
ramétrer et personnaliser Calc, pour plus de détails.

